# Quantization of the Lorentz Force

Pascal Szriftgiser[1] and Edgardo S. Cheb-Terrab[2]
(1) Laboratoire PhLAM, UMR CNRS 8523, Université Lille 1, F-59655, France
(2) Maplesoft

We consider the case of a quantum, non-relativistic, particle with mass $m$ and charge $q$ evolving under the action of an arbitrary time-independent magnetic field $\vec{B} = \nabla \times \vec{A}$, where $\vec{A}$ is the vector potential. The Hamiltonian for this system is

$$H = \frac{\left(\vec{p} - q\,\vec{A}\right)^2}{2\,m}$$

where $\vec{p}$ is the momentum of the particle, and the force acting in this particle, also called the *Lorentz* force, is given by

$$\vec{F} = m\,\dot{\vec{v}}$$

where $\vec{v}$ is the quantized velocity of the particle, and all of $H, \vec{p}, \vec{v}, \vec{B}, \vec{A}$ and $\vec{F}$ are Hermitian quantum operators representing observable quantities.

In the *classic (non-quantum) case*, the Lorentz force $\vec{F}$ for such a particle in the absence of electrical field is given by

$$\vec{F} = q\,\vec{v} \times \vec{B},$$

**Problem:** Departing from the Hamiltonian, show that in the *quantum case* the Lorentz force is given by [1]

$$\vec{F} = \frac{q\left(\vec{v} \times \vec{B} - \vec{B} \times \vec{v}\right)}{2}$$

[1] Photons et atomes, Introduction à l'électrodynamique quantique, p. 179, Claude Cohen-Tannoudji, Jacques Dupont-Roc et Gilbert Grynberg - EDP Sciences janvier 1987.

**Solution**

We choose to tackle the problem in Heisenberg's picture of quantum mechanics, where the state of a system is static and only the quantum operators evolve in time according to

$$\dot{\mathrm{O}}(t) = \frac{\mathrm{i}}{\hbar} \cdot [H, \mathrm{O}(t)]_{-}$$

Also, the algebraic manipulations are simpler using tensor abstract notation instead of the standard 3D vector notation. We then start setting the framework for the problem, a system of coordinates X, indicating the dimension of the tensor space to be 3 and the metric Euclidean, and that we will use lowercaselatin letters to represent tensor indices. In addition, not necessary but for convenience, we set the lowercase latin *i* to represent the imaginary unit and we request *automaticsimplification* so that the output of everything comes automatically simplified in size.

```
> restart; with(Physics) : interface(imaginaryunit = i) :
> Setup(mathematicalnotation = true, automaticsimplification = true, coordinates = X, dimension
    = 3, metric = Euclidean, spacetimeindices = lowercaselatin, quiet)
```

$$[automaticsimplification = true, coordinatesystems = \{X\}, dimension = 3, mathematicalnotation = true, metric = \{(1, 1) = 1, (2, 2) = 1, (3, 3) = 1\}, spacetimeindices = lowercaselatin] \quad (1)$$

Next we indicate the letters we will use to represent the quantum operators with which we will work, and also the standard commutation rules between position and momentum, always the starting point when dealing with quantum mechanics problems

```
> Setup(quantumoperators = {F}, hermitianoperators = {r, x, p, v, A, B, H}, realobjects = {ℏ, m,
    q}, algebrarules = {%Commutator(x[k], x[l]) = 0, %Commutator(p[k], p[n]) = 0,
    %Commutator(x[k], p[l]) = i·ℏ·KroneckerDelta[k, l]})
```

$$\left[algebrarules = \left\{ [p_k, p_n]_- = 0, [x_k, p_l]_- = i\,\hbar\,\delta_{k,l}, [x_k, x_l]_- = 0 \right\}, hermitianoperators = \{A, B, H, p, r, v, x\}, quantumoperators = \{A, B, F, H, p, r, v, x\}, realobjects = \{\hbar, m, q, x1, x2, x3, \square, \square\}\right] \quad (2)$$

Note that we start *not indicating* $F$ as Hermitian, *in order to arrive* at that result. The quantum operators $A$, $B$, and $F$ are explicit functions of X, so to avoid redundant display of this functionality on the screen we use

```
> CompactDisplay((A, B, F)(X))
```

*A(x1, x2, x3) will now be displayed as A*

*B(x1, x2, x3) will now be displayed as B*

*F(x1, x2, x3) will now be displayed as F* **(3)**

Define now as *tensors* the quantum operators that we will use with tensorial notation (recalling: for these, Einstein's sum rule for repeated indices will be automatically applied when simplifying)

```
> Define(x, p, v, A, B, F, quiet)
```

$$\{A, B, F, p, v, x, \gamma_a, \sigma_a, X_a, \partial_a, g_{a,b}, \delta_{a,b}, \epsilon_{a,b,c}\} \quad (4)$$

The Hamiltonian,

$$H = \frac{(\vec{p} - q\,\vec{A})^2}{2\,m}$$

in tensorial notation, is given by

$$> H = \frac{1}{2\,m}\,(p[n] - q\,A[n](X))^2$$

$$H = \frac{(p_n - q\,A_n)^2}{2\,m} \quad (5)$$

Generally speaking to arrive at $\vec{F} = \frac{q\,(\vec{v} \times \vec{B} - \vec{B} \times \vec{v})}{2}$ what we now need to do is

1) Express this Hamiltonian **(5)** in terms of the velocity

And, recalling that, in Heisenberg's picture, quantum operators evolve in time according to

$$\dot{\mathrm{O}}(t) = \frac{\mathrm{i}}{\hbar} \cdot [H, \mathrm{O}(t)]_{-}$$

2) Take the commutator of $H$ with the velocity itself to obtain its time derivative and, from $\vec{F} = m\,\dot{v}$, that commutator is already the force up to some constant factors.

To get in contact with the basic commutation rules between position and momentum behind quantum phenomena, the quantized velocity itself can be computed as the time derivative of the position operator, i.e as the commutator of $x_k$ with $H$

> $\frac{\mathrm{i}}{\hbar} \cdot Commutator(\mathbf{(5)}, x[k])$

$$\frac{\mathrm{i}\,[H, x_k]_{-}}{\hbar} = \frac{\mathrm{i}\,q^2\,\left[A_n, [A_n, x_k]_{-}\right]_{+} - \mathrm{i}\,q\,\left[p_n, [A_n, x_k]_{-}\right]_{+} - 2\,(q\,A_n - p_n)\,\delta_{k,n}\,\hbar}{2\,\hbar\,m} \quad \mathbf{(6)}$$

This expression for the velocity, that involves commutators between the potential $A_n$, the position $x_k$ and the momentum $p_n$, can be simplified taking into account the basic quantum algebra rules between position and momentum. We assume that $A_n$(X) can be decomposed into a formal power series (possibly infinite) of the $x_k$, hence all the $A_n$ commute between themselves as well as with all the $x_k$ :

> $\{\%Commutator(A[k](X), x[l]) = 0, \%Commutator(A[k](X), A[l](X)) = 0\}$

$$\left\{[A_k, x_l]_{-} = 0, [A_k, A_l]_{-} = 0\right\} \quad \mathbf{(7)}$$

(Note: in some cases, this is not true, but those cases are beyond the scope of this worksheet.)

Add these rules to the algebra rules already set so that they are all taken into account when simplifying things

> $Setup(algebrarules = \mathbf{(7)})$

$$\left[algebrarules = \left\{[p_k, p_n]_{-} = 0, [x_k, p_l]_{-} = \mathrm{i}\,\hbar\,\delta_{k,l}, [x_k, x_l]_{-} = 0, [A_k, x_l]_{-} = 0, [A_k, A_l]_{-} = 0\right\}\right] \quad \mathbf{(8)}$$

> $Simplify(\mathbf{(6)})$

$$\frac{\mathrm{i}\,[H, x_k]_{-}}{\hbar} = \frac{-A_k\,q + p_k}{m} \quad \mathbf{(9)}$$

The right-hand side of **(9)** is then the $k^{th}$ component of the velocity tensor quantum operator, the relationship is the same as in the classical case

> $v[k] = rhs(\mathbf{(9)})$

**(10)**

$$v_k = \frac{-A_k\, q + p_k}{m} \tag{10}$$

and with this the Hamiltonian **(5)** can now be rewritten in term of the velocity completing step 1)

> *simplify*(**(5)**, {*SubstituteTensorIndices*(*k* = *n*, (*rhs* = *lhs*)(**(10)**))})

$$H = \frac{m\, v_n^{\;2}}{2} \tag{11}$$

For step 2), to compute

$$\vec{F} = m\,\dot{v} = \frac{\mathrm{i}\, m\, [H, v_k]_-}{\hbar}$$

we need the commutator between the different components of the quantized velocity which, contrary to what happens in the classical case, do not commute. For this purpose, take the commutator between **(10)** with itself after replacing the free index

> *Commutator*(**(10)**, *SubstituteTensorIndices*(*k* = *n*, **(10)**))

$$[v_k, v_n]_- = -\frac{q\,\left([A_k, p_n]_- + [p_k, A_n]_-\right)}{m^2} \tag{12}$$

To simplify **(12)**, we use the fact that if $f$ is a commutative mapping that can be decomposed into a formal power series in all the complex plan (which is assumed to be the case for all $A_k$(X)), then

$$[p_k, f(x, y, z)]_- = -\mathrm{i}\,\hbar\,\left(\partial_k(f(x, y, z))\right)$$

where $p_k = -i\,\hbar\,\partial_k$ is the momentum operator along the $x_k$ axis. This relation reads in tensor notation:

> *Commutator*(*p*[*k*], *A*[*n*](*X*)) = -i·ℏ·*d_*[*k*](*A*[*n*](*X*))

$$[p_k, A_n]_- = -\mathrm{i}\,\hbar\,\left(\partial_k(A_n)\right) \tag{13}$$

Add this rule to the rules previously set in order to automatically take it into account in **(12)**

> *Setup*(**(13)**)

$$\left[algebrarules = \left\{[p_k, p_n]_- = 0,\ [p_k, A_n]_- = -\mathrm{i}\,\hbar\,\left(\partial_k(A_n)\right),\ [x_k, p_l]_- = \mathrm{i}\,\hbar\,\delta_{k,\,l},\ [x_k, x_l]_- = 0,\ [A_k, x_l]_- = 0,\ [A_k, A_l]_- = 0\right\}\right] \tag{14}$$

> **(12)**

$$[v_k, v_n]_- = \frac{-\mathrm{i}\,q\,\hbar\,\left(\partial_n(A_k) - \left(\partial_k(A_n)\right)\right)}{m^2} \tag{15}$$

Also add this other rule so that it is taken into account automatically

> *Setup*(**(15)**)

$$\left[algebrarules = \left\{[p_k, p_n]_- = 0,\ [p_k, A_n]_- = -\mathrm{i}\,\hbar\,\left(\partial_k(A_n)\right),\ [v_k, v_n]_- = \frac{-\mathrm{i}\,q\,\hbar\,\left(\partial_n(A_k) - \left(\partial_k(A_n)\right)\right)}{m^2},\ [x_k, p_l]_- = \mathrm{i}\,\hbar\,\delta_{k,\,l},\ [x_k, x_l]_- = 0,\ [A_k, x_l]_- = 0,\ [A_k, A_l]_- = 0\right\}\right] \tag{16}$$

Recalling now the expression of the Hamiltonian **(11)** as a function of the velocity, one can compute the components of the force operator $\left(\vec{F}\right)_k = m\,\dot{v}_k = \frac{\mathrm{i}\,m\,[H, v_k]_-}{\hbar}$

> $F[k](X) = m\cdot\frac{\mathrm{i}}{\hbar}\cdot\%Commutator(rhs((\mathbf{11})), v[k])$

$$F_k = \frac{\mathrm{i}\,m\left[\frac{m\,v_n^{\,2}}{2}, v_k\right]_-}{\hbar} \qquad \mathbf{(17)}$$

Simplify this expression for the quantized force taking the quantum algebra rules **(16)** into account

> $Simplify((\mathbf{17}))$

$$F_k = \frac{q\left(-\left(\partial_n(A_k)\right)v_n + \left(\partial_k(A_n)\right)v_n - v_n\left(\partial_n(A_k)\right) + v_n\left(\partial_k(A_n)\right)\right)}{2} \qquad \mathbf{(18)}$$

It is not difficult to verify that this is the antisymmetrized vector product $\vec{v}\times\vec{B}$. Departing from $\vec{B} = \nabla\times\vec{A}$ expressed using tensor notation,

> $B[c](X) = LeviCivita[c, n, m]\cdot d\_[n](A[m](X))$

$$B_c = -\epsilon_{c, m, n}\left(\partial_n(A_m)\right) \qquad \mathbf{(19)}$$

and taking into acount that

$$\left(\vec{v}\times\vec{B}\right)_k = \epsilon_{b, c, k}\,v_b\,B_c$$

multiply both sides of **(19)** by $\epsilon_{b, c, k}\,v_b$, getting

> $LeviCivita[k, b, c]\,v[b]\cdot(\mathbf{19})$

$$\epsilon_{b, c, k}\,v_b\,B_c = -\epsilon_{b, c, k}\,\epsilon_{c, m, n}\,v_b\left(\partial_n(A_m)\right) \qquad \mathbf{(20)}$$

> $Simplify((\mathbf{20}))$

$$\epsilon_{b, c, k}\,v_b\,B_c = v_m\left(\partial_k(A_m)\right) - v_n\left(\partial_n(A_k)\right) \qquad \mathbf{(21)}$$

Finally, replacing the repeated index *m* by *n*

> $SubstituteTensorIndices(m = n, (\mathbf{21}))$

$$\epsilon_{b, c, k}\,v_b\,B_c = v_n\left(\partial_k(A_n)\right) - v_n\left(\partial_n(A_k)\right) \qquad \mathbf{(22)}$$

Likewise, for

$$\left(\vec{B}\times\vec{v}\right)_k = \epsilon_{b, c, k}\,B_b\,v_c$$

multiplying **(19)**, this time from the right instead of from the left, we get

> $Simplify((\mathbf{19})\cdot LeviCivita[k, b, c]\cdot v[b])$

$$\epsilon_{b, c, k}\,B_c\,v_b = \left(\partial_k(A_m)\right)v_m - \left(\partial_n(A_k)\right)v_n \qquad \mathbf{(23)}$$

> $SubstituteTensorIndices(m = n, (\mathbf{23}))$

$$\epsilon_{b, c, k}\,B_c\,v_b = \left(\partial_k(A_n)\right)v_n - \left(\partial_n(A_k)\right)v_n \qquad \mathbf{(24)}$$

Simplifying now the expression **(18)** for the quantized force taking into account **(22)** and **(24)** we get

> $simplify((\mathbf{18}), \{(rhs = lhs)((\mathbf{22})), (rhs = lhs)((\mathbf{24}))\})$

$$F_k = \frac{q\,\epsilon_{b, c, k}\left(v_b\,B_c + B_c\,v_b\right)}{2} \qquad \mathbf{(25)}$$

i.e.

$$\vec{F} = \frac{q\,(\vec{v} \times \vec{B} - \vec{B} \times \vec{v})}{2}$$

in tensor notation. Finally, we note that this operator is Hermitian as expected

> **(25)** − *Dagger*(**(25)**)

$$F_k - F_k^{\dagger} = 0 \qquad \textbf{(26)}$$